



THE
POWER
TO KNOW.

Technical Paper

Changing language during a SAS session

Table of Contents

| | |
|---|-----------|
| Abstract | 1 |
| Introduction | 1 |
| Configuration | 1 |
| Sample scenarios | 1 |
| Scenario 1: Connect with SAS® Data Integration Studio to the SAS Unicode Server.. | 2 |
| Scenario 2: Create a bilingual report | 3 |
| Scenario 3: Create separate reports depending on the language..... | 4 |
| Restrictions | 7 |
| Conclusion | 7 |
| Acknowledgments | 8 |
| References | 8 |
| Appendix: Setting up language switching in legacy SAS environments | 9 |
| Copy the templates | 9 |
| Special considerations | 10 |
| Create a new configuration file | 10 |
| Glossary | 12 |

Content provider for Changing Language during a SAS Session was
Manfred Kiefer, Globalization Specialist, SAS Heidelberg.

Abstract

SAS[®] products have been localized, or translated into, since SAS 6. *Localization* of these products involves a translation of message files, ODS templates, the SAS registry, and other files. The language used for messages displayed by SAS is determined at start up by settings in the SAS configuration file. In SAS 9.2, support was added to the Unicode server so that the language of certain texts matches the SAS LOCALE= option when localization is available.

This feature, referred to as *Language Switching*, is controlled by the LOCALELANGCHG option. If LOCALELANGCHG is on, the language switching feature is enabled and the setting of the LOCALE= option determines the language used for procedure output and user interface elements. ODS fonts used by a product or procedure also match the LOCALE. Messages written to the SAS log, including notes, warnings, and errors, always display in the language used at SAS system start up.

Introduction

People should be able to do their job without having to worry about the language. This is why SAS software is available in many languages. The SAS[®] Unicode server even goes a step further by allowing you to switch the language on the fly.

Before we go into more detail, let us have a closer look at configuration of the SAS Unicode server.

Configuration

How to install and configure a SAS Unicode Server has been explained in the Technical Paper *Processing Multilingual Data with the SAS[®] 9.2 Unicode Server*. By default, a SAS Unicode Server session in 9.2 has set the option LOCALELANGCHG. If you want to switch this option off, you need to add the following line to the SAS configuration file which is located in !SASROOT/nls/u8:
-NOLOCALELANGCHG.

Sample scenarios


Being able to switch the language of reports on the fly is ideal for countries that have more than one official language. Belgium, for example, has Dutch, French and German, Finland Finnish and Swedish, and Switzerland has four national languages: German, French, Italian, and Rhaeto-Romanic. In Japan English is the foremost foreign language studied and taught. So publications might appear in Japanese and in English. In all these cases, you can use various client applications in conjunction with a SAS Unicode Server session to produce multilingual reports on the fly. Below are some sample scenarios to illustrate the process.

Scenario 1: Connect with SAS® Data Integration Studio to the SAS Unicode Server

A simple test can be run by submitting some code to the server to run a Natural Language (NL) format. The NL formats are locale-sensitive and should return values that reflect the locale being set. See the SAS 9.2 *National Language Support Reference Guide* for details on using the NL formats. In our example we also include a PROC OPTIONS step to verify the option values:

```
proc options group=languagecontrol; run;
data _null_; d=today(); put d nldate.; run;
```

In this case we have set the locale for SAS® Data Integration Studio to French by using the Locale Setup Manager:



The result of our simple test program looks like this:

The screenshot shows the SAS Data Integration Studio interface. The main window displays the SAS log output for a job named 'New Job 97054'. The log includes the following text:

```

FSDETYPE=DEFAULT Specifies a full-screen double-byte character set (DBCS) encoding method
FSIMM= Specifies a full-screen double-byte character set (DBCS) input method module
FSIMMOPT= Specifies options for full-screen double-byte character set (DBCS) input method module
ENCODING=UTF-8 Specifies default encoding for internal processing of data.
LOCALE=FR_FR Specifies the current locale for the SAS session.
NONLSCOMPATMODE Uses the user specified encoding to process character data
NOTE: PROCEDURE _DISARM_ STOP_ _DISARM_ 2008-09-25T04:18:31.752-04:00| _DISARM_ WorkspaceSe
rver| _DISARM_ SAS| _DISARM_ |
         _DISARM_ 631904| _DISARM_ 6328320| _DISARM_ 11| _DISARM_ 11| _DISARM_ 9| _DISARM_ 1166| _DISAR
M| 0.000000| _DISARM_ 0.076500|
         _DISARM_ 1537949911.675553| _DISARM_ 1537949911.752053| _DISARM_ 0.076500| _DISARM_ | _ENDDISA
RM
NOTE: PROCEDURE OPTIONS used (Total process time):
      real time      0.08 seconds
      cpu time       0.00 seconds

85      data _null_; d=today(); put d nldate.; run;

25 septembre 2008
NOTE: PROCEDURE _DISARM_ STOP_ _DISARM_ 2008-09-25T04:18:32.311-04:00| _DISARM_ WorkspaceSe

```

The 'Propriétés de base' window shows the following details for 'New Job 97054':

| Nom | Valeur |
|--------------------|---------------------|
| Nom | New Job 97054 |
| Description | My first test jo... |
| Emplacement d... | /Users/sasdem... |
| Extrait par | |
| Tables chargées | |
| Dernière modif.... | sasdemo |

The 'Gestionnaire d'exécution' window shows the following execution details:

| # | Job | Etat | Heure de début | Heure de fin | Serveur d'applications utilisé |
|---|---------------|---------|------------------------|------------------------|--------------------------------|
| 1 | New Job 97054 | Terminé | 25 sept. 2008 04:18:30 | 25 sept. 2008 04:18:32 | SASApp |

We can see that the Unicode Server session has picked up the language of the client. The date is also written out in the current locale.

Scenario 2: Create a bilingual report

Here we are using SAS® Enterprise Guide® and the SAS Unicode Server to create a report in both English and Japanese. We are switching the language on the fly:


```
ods html file='bilingual-report.html';
```

```
options locale=en_US;
title "executed in &sysvlong by &sysuserid on machine &SYSTCPIPHOSTNAME";
footnote "%sysfunc(datetime(),nldatm.)";
proc univariate data=sashelp.class;
var age;
run;
```

```
options locale=ja_JP;
title "マシン &SYSTCPIPHOSTNAME.で、 &sysuserid によって &sysvlong で実行されました。";
footnote %sysfunc(datetime(),nldatm.);
proc univariate data=sashelp.class;
```

```
var age;
run;
ods html close;
```

This is what the result looks like:



Scenario 3: Create separate reports depending on the language

In this example we are assuming that we create lists of Swiss customers from a multilingual database. Depending on the users' preferred language we create separate lists for French, German and Italian-speaking customers. For output we sort these lists according to local language conventions using SORTSEQ=LINGUISTIC. Using this option produces a collating sequence that is culturally correct. More information on linguistic collation can be obtained from the SAS technical paper *Linguistic Collation: Everyone Can Get What They Expect - Sensible Sorting for Global Business Success*.

Here's the code:

```
/* -----
Create subsets of the data
----- */

proc sql;
create table french as
select name, first, street, zip, city from temp.swiss where lang = 'fra';
4
```



```

create table german as
select name, first, street, zip, city from temp.swiss where lang = 'deu';
create table italian as
select name, first, street, zip, city from temp.swiss where lang = 'ita';
quit;

/* -----
   label variables in different languages
   ----- */
data french;
  set french;
  label name='Nom';
  label first='Prénom';
  label street='Adresse';
  label city='Lieu';
  label zip='CP';
run;

data german;
  set german;
  label name='Name';
  label first='Vorname';
  label street='Strasse';
  label city='Ort';
  label zip='PLZ';
run;

data italian;
  set italian;
  label name='Nome';
  label first='Cognome';
  label street='Strada';
  label city='Città';
  label zip='CAP';
run;
/* -----
   Run the SORT Procedure and create RTF output
   ----- */
ods rtf file='c:\temp\french.rtf';
options locale=fr_CH dflang=locale;
TITLE;
TITLE1 "Liste des clients";
TITLE2 "par ordre alphabétique";
FOOTNOTE;
FOOTNOTE1 "Générée par le système SAS le %TRIM(%SYSFUNC(DATE(), NLDATE20.)) à
%SYSFUNC(TIME(), NLTIMAP10.)";

proc sort data=french sortseq=linguistic;
  by name;
run;
proc print data=french label noobs; run;

RUN; QUIT;

TITLE; FOOTNOTE;
/* -----
   DTRESET specifies that SAS update the date and time in the titles
   ----- */

```

```

of the SAS log and the listing file.
----- */
OPTIONS DTRESET;

ods rtf file='c:\temp\german.rtf';
options locale=de_CH dflang=locale;
TITLE;
TITLE1 "Kundenliste";
TITLE2 "alphabetisch sortiert";
FOOTNOTE;
FOOTNOTE1 "Generiert durch das SAS System am %TRIM(%QSYSFUNC(
DATE()), NLDATE20.)) um %SYSFUNC(TIME()), NLTIMAP10.)";

proc sort data=german sortseq=linguistic;
  by name;
run;
proc print data=german label noobs; run;

RUN; QUIT;
TITLE; FOOTNOTE;

/* -----
DTRESET specifies that SAS update the date and time in the titles
of the SAS log and the listing file.
----- */
OPTIONS DTRESET;

ods rtf file='c:\temp\italian.rtf';
options locale=it_CH dflang=locale;
TITLE;
TITLE1 "Lista dei clienti";
TITLE2 "in ordine alfabetico";
FOOTNOTE;
FOOTNOTE1 "Generato dal sistema SAS il %TRIM(%QSYSFUNC(
DATE()), NLDATE20.)) alle %SYSFUNC(TIME()), NLTIMAP10.)";

proc sort data=italian sortseq=linguistic;
  by name;
run;
proc print data=italian label noobs; run;

RUN; QUIT;
TITLE; FOOTNOTE;
ods rtf close;

```

And this is what the list of French-speaking customers looks like:

lundi, 13. octobre 2008 07.

*Liste des clients
par ordre alphabétique*

| Nom | Prénom | Adresse | CP | Lieu |
|-----------|-----------|-----------------------------|------|----------------------|
| Boucher | Corinne | Rte du Lézard 29 | 1663 | Moléson-sur-Gruyères |
| Côté | Frédéric | Rte du Baly 9 | 1636 | Broc |
| Cotée | Madeleine | Rue des Grand-Chênes 4 | 1752 | Cormanon |
| Dubois | Benoît | rue des fahys 143 21/2 | 2000 | Neuchâtel |
| Fournier | Étienne | Route des Mords 29 | 1632 | Riaz |
| Girard | Régine | Rue des Corbaz 7 | 1092 | Belmont-sur-Lausanne |
| Legrand | Claire | 2, chemin du Tonkin | 1242 | Satigny |
| Martin | Désirée | chemins des chauxfours 37 a | 2802 | Develier |
| Thibeault | Georges | Rue de la prairie, 25 | 2854 | Bassecourt |
| Vaudron | Sébastien | rue de la Chocolaterie 30 | 1026 | Echandens |

Restrictions

There are a few restrictions:

1. Language Switching only works with the SAS Unicode server (but see the Appendix below).
2. Messages written to the SAS log, including notes, warnings, and errors, always display in the language used at SAS system start up. Usually this is English.
3. At the time of appearance of this publication not all clients have full support for Language Switching. SAS[®] Web Report Studio and SAS[®] Enterprise Guide[®], for example, do support this feature, but other BI clients may, or may not support this. Check with your local SAS office and or technical support for more information.

Conclusion

Language switching with the SAS Unicode server, allows you to produce multilingual output in one SAS session.

Acknowledgments

The author wishes to thank Betty Joyner and Wu Deng for reviewing this paper. A special thank you to Elizabeth Bales and Joji Kobayashi for providing the details on how to switch language in legacy environments, and to Rachel Wynne for providing editorial assistance. Finally, I would also like to thank those employees who have contributed to the effort to implement the LOCALELANGCHG option within the SAS System.

References

SAS Institute Inc. 2008. SAS® 9.2 National Language Support (NLS): Reference Guide. Cary, NC: SAS Institute Inc.

SAS Institute Inc. 2008. SAS Institute technical paper. "Linguistic Collation: Everyone Can Get What they Expect - Sensible Sorting for Global Business Success."

SAS Institute Inc. 2008. SAS Institute technical paper. "Processing Multilingual Data with the SAS® 9.2 Unicode Server."

Appendix: Setting up language switching in legacy SAS environments

The LOCALELANGCHG option which controls the feature can be turned on in legacy systems, those with a session encoding other than UTF-8. However, the feature is not officially supported in those environments for 9.2.

If a customer wishes to use Language Switching in a legacy SAS environment in 9.2 some manual setup is required after SAS is installed. This section explains what the customer has to do in order to enable Language Switching.

Note that the examples provided are written for SAS installed on Windows. The steps for UNIX are very similar. Of course, the command syntax you need to use is different.

Copy the templates

Support for language switching requires a special set of template itemstore files. These itemstores can be 'created' in the SAS install location by copying installed template itemstore files to a new directory and renaming them using a special naming convention. The new itemstore files should be named `tmpl_xx_.sas7bitm`, where 'xx' is the 2-character language code used by SAS. All of the installed template itemstores (`tmplmst.sas7bitm`) must be copied, including the English and translated files. The steps to follow are below.

Create a new directory for the copied itemstores. Placing the copied files in a separate directory prevent any confusion with other files installed with the system. For example, create a new directory under `!SASROOT\nls\en\sashelp` for these files.

Copy all of the translated template itemstores from the language subdirectories under `!SASROOT\nls` to the new directory (`!SASROOT\nls\en\sashelp`). The new copy should be named `tmpl_xx_.sas7bitm`, where 'xx' is the 2-character language code.

Copy the English template itemstore from the SASHELP directory to the new directory created in step 1. The copy should be named `tmpl_en_.sas7bitm`. For example, on Windows SASHELP files are installed in `!SASROOT\core\sashelp`.

On Windows, if SAS 9.2 is installed under `C:\Program Files\SAS`, the following code can be used to create the new directory and copy the files:

```
set SASROOT=C:\Program Files\SAS\SASFoundation\9.2
set NLSROOT="%SASROOT%\nls
set ENHELP="%NLSROOT%\en\sashelp
mkdir %ENHELP%;
copy "%NLSROOT%\cs\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_cs_.sas7bitm
copy "%NLSROOT%\de\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_de_.sas7bitm
copy "%NLSROOT%\es\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_es_.sas7bitm
copy "%NLSROOT%\fr\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_fr_.sas7bitm
copy "%NLSROOT%\hu\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_hu_.sas7bitm
copy "%NLSROOT%\it\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_it_.sas7bitm
copy "%NLSROOT%\pl\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_pl_.sas7bitm
copy "%NLSROOT%\ru\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_ru_.sas7bitm
```

```
copy "%NLSROOT%"\sv\sashelp\tmplmst.sas7bitm ENHELP%\tmpl_sv_.sas7bitm
copy "%SASROOT%"\core\sashelp\tmplmst.sas7bitm %ENHELP%\tmpl_en_.sas7bitm
```

Special considerations

In order to successfully use a template itemstore for language switching in a legacy environment, it must have been created using an encoding that matches the SAS session encoding. In other words, language switching in a legacy environment only makes sense with languages that use the same encoding. For example, Western European languages such as French, German, Italian, Spanish, and Swedish all share the Latin1 character repertoire; Central and Eastern European languages such as Czech, Hungarian and Polish share the Latin2 character repertoire. Hence in a legacy environment, you can switch between French and German, for example, or between Czech and Polish but not between French and Polish. Switching between Asian (CJK) languages in a legacy environment does not make sense at all.

When the LOCALE option is set, if the LOCALE and ENCODING are not compatible the following is true:

- ODS PATH is not updated. The ODS PATH that was in effect before the LOCALE option was run and remains.
- SAS messages do not use the language of the current locale. Instead, the language used to start the SAS session is used.

Therefore, if the language of the previous locale is not the same as the language of the startup locale, procedure output could contain mixed languages.

Create a new configuration file

The SAS configuration file, called sasv9.cfg, needs to be updated to turn on the Language Switching feature. The SASHELP option in the configuration file must also be updated to include the directory created for the copied itemstore files. We recommend making a copy of sasv9.cfg file to use specifically for language switching support.

Several copies of sasv9.cfg are installed for SAS. The master is in the !SASROOT directory. Each localization installed for SAS also has a configuration file. These configuration files are located in a directory named with the 2-character language name that is under !SASROOT\nls. Select the sasv9.cfg file for the language that is used when the SAS session starts. For example, to run SAS on Windows using the French localizations, copy of !SASROOT\nls\fr\sasv9.cfg and make the changes below to the copied configuration file.

1. Set the LOCALELANGCHG option by adding this line:

```
-LOCALELANGCHG
```

2. Update the SASHELP option to add the directory created for the copied template itemstore files. For example, SASHELP option in the French configuration file on Windows might look like this:

```
/* Setup the SAS System help directory definition*/
-SASHELP      (
                "!SASCFG\SASCFG"
                "!sasext1\fr\sashelp"
                "!sasext1\core\sashelp"
                "!sasext1\nls\en\sashelp" )
```

Once the new configuration file has been updated, it must be picked up by the SAS command. To find out more about updating and adding configuration files, consult the SAS 9.2 Companion for the operating environment where SAS is installed.

Glossary

CJK

This is a collective term for Chinese, Japanese, and Korean. The term is used in the field of software localization and internationalization.

Locale

A value that reflects the language, local conventions, and culture for a geographic region. Local conventions can include specific formatting rules for dates, times, and numbers, and a currency symbol for the country or region. Collating sequences, paper sizes, and conventions for postal addresses and telephone numbers are also typically specified for each locale. Some examples of locale values are `French_Canada`, `Portuguese_Brazil`, and `English_USA`.

Localization

The process of adapting a product to meet the language, cultural, and other requirements of a specific target environment or market so that customers can use their own languages and conventions when using the product. Translation of the user interface, system messages, and documentation is a large part of the localization process.

Unicode

A 16-bit encoding that supports the interchange, processing, and display of characters and symbols from dozens of writing systems, for a total of up to 65,536 characters. Unicode includes all characters from almost all modern written languages as well as characters from some historical languages.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2008 SAS Institute Inc., Cary, NC, USA. All rights reserved.